
Exceções

1. Manipulando Exceções

Quando um erro ocorre no aplicativo durante a execução, a aplicação chama uma exceção . Uma exceção é um objeto que contém informações sobre o erro ocorrido e onde ocorreu. Se  não for especificado no código uma resposta ao erro, o Delphi mostrará uma mensagem descrevendo o erro : 

No exemplo, poderia-se clicar no botão OK e continuar a executar a aplicação. Mas muitas vezes o erro ocorrido impede a continuação da operação, ou leva a perda de informações valiosas.

O Ideal seria que se pudesse tratar os erros ocorridos, evitando a perda de dados ou a necessidade de encerrar a aplicação. Além de tratar o erro , a rotina de tratamento de erros poderia enviar ao usuário uma mensagem em português , mais  significativa : 

O Delphi permite que se desenvolva aplicações robustas, ou seja, que tratam os erros ocorridos de forma que os dados não se percam . Para fazer uma aplicação robusta, é necessário que o código reconheça as excessões quando elas aparecerem, respondendo a elas.

Cria-se uma resposta a uma exceção em um bloco de código, chamado de bloco protegido porque está resguardado de erros que podem parar a aplicação ou perder dados. Um bloco protegido começa com a palavra reservada try e termina com a palavra reservada end. Entre essas palavras determina-se os comandos protegidos e a forma de reação aos erros.

Quando se define um bloco protegido, especifica-se respostas a exceções que podem ocorrer dentro deste bloco. Se a exceção ocorrer, o fluxo do programa pula para a resposta definida, e após executá-la, abandona o bloco.

A forma mais simples para responder a uma exceção é garantir que algum código limpo é executado. Este tipo de resposta não corrige o erro, mas garante que sua aplicação não termine de forma instável. Normalmente, usa-se este tipo de resposta para garantir a liberação de recursos alocados, mesmo que ocorra um erro.

Outra forma de se responder a uma exceção é tratando o erro , ou seja, oferecendo uma resposta específica para um erro específico. Ao tratar o erro destrui-se a exceção , permitindo que a aplicação continue a rodar.

2. A Estrutura Try..Finally..End
O código precisa garantir que , mesmo ocorrendo um erro, os recursos alocados sejam desalocados. Entre estes recursos estão : arquivos, memória, recursos do windows, objetos.

Para garantir a desalocação dos recursos , usamos a estrutura abaixo :

{aloca os recursos}

try

      {comandos que usam os recursos}

finally
      { libera os recursos }

end;
A aplicação sempre executará os comandos inseridos na parte finally do bloco, mesmo que uma exceção ocorra. Quando um erro ocorre no bloco protegido, o programa pula para a parte finally , chamada de código limpo, que é executado. Mesmo que não ocorra um erro, estes comandos são executados.

No código a seguir, foi alocada memória e gerado um erro, ao tentar-se a divisão por 0 (zero). Apesar do erro,  o programa libera a memória alocada:

Procedure Tform1.Button1click (Sender : Tcomponent );

Var
Ponteiro : Pointer;

Inteiro, Dividendo : Integer;

Begin
        Dividendo:= 0;

        GetMem(Ponteiro, 1024);

        Try
                 Inteiro: = 10 div dividendo;

         Finally

   FreeMem(Ponteiro, 1024);

          End;

End;

3. Exceções da RTL
Quando se escreve código chama rotinas da biblioteca de run-time ( RTL, run-time library), como funções matemáticas ou de manipulação de arquivos, os erros aparecem na forma de exceções. Por padrão, a RTL manda uma mensagem para o usuário, mas pode-se tratar estes erros de outra forma.

As exceções geradas pela RTL são definidas na unit  SysUtils , e todas descendem da classe mais geral Exception, que provê a string que aparece no quando de mensagem da exceção.

Há sete tipos de exceções geradas pela RTL :

· Exceções de entrada e saída;

· Exceções de memória heap;

· Exceções matemáticas envolvendo inteiros;

· Exceções matemáticas envolvendo pontos flutuantes;

· Exceções de typecast;

· Exceções de conversão;

· Exceções de Hardware;

· Exceções de Entrada e Saída
Ocorrem quando RTL tenta acessar arquivos ou dispositivos de entrada e saída.

A Unit  SysUtils define uma exceção genérica de entrada e saída chama  EinOutError, que contém um atributo chamado ErrorCode que indica o erro ocorrido.

· Exceções de Memória Heap
Ocorrem quando se tenta alocar ou acessar memória dinâmica. São definidos dois tipos de exceção : EoutofMemory (indica que não há memória suficiente) e EinvalidePointer (Ponteiro Inválido).

· Exceções Matemáticas para Inteiros
Ocorrem quando se realiza operações com números inteiros. A Unit  SysUtils define uma exceção genérica chamada EintError, com três tipos de exceção derivadas : EdividByZero (Divisão por zero), ErangeError (Número fora da região), e EintOverFlow (Estouro na operação com inteiro).

· Exceções de Pontos Flutuantes
Ocorrem quando se realiza operações com dados do tipo real. A Unit  SysUtils define uma exceção genérica chamada EMathError, com as seguintes  exceções derivadas : EinvalidOp ( Processador encontrou uma instrução indefinida),  EZeroDivide (Divisão por zero), EOverFlow ,  EUnderFlow.

· Exceções de TypeCast
Ocorrem quando você tenta atribuir um objeto a um tipo inválido usando o operados as . Gera-se a exceção  EinvalidCast quando isto ocorre.

· Exceções de Conversão
Ocorrem quando se convertem dados de um tipo para outro. A Unit SysUtils define uma exceção chamada  EconvertError quando isto ocorre, avisando que a conversão não pode ser feita.

· Exceções de Hardware
Podem ocorrer em dois tipos de situação: quando o processador detecta uma falha, ou quando a aplicação gera uma interrupção intencional.

A Unit SysUtils define uma exceção genérica chamada  EprocessorException, com as exceções derivada : Efault (Exceção Base), EGPFault (Falha Geral de Proteção),  EstackedFault (acesso ilegal ao seguimento stack do processador),  EpageFault (o gerenciador de memória do windows não está conseguindo realizar o qrquivo de swap), EinvalidOpCode ( o processador encontra uma instrução indefinida), EbreakPoint  (a aplicação gerou uma interrupção breakpoint),  Esinglestep (a aplicação gera uma interrupção single-step).

4. A Estrutura Try..Except..End
Um tratamento de exceção é um código que trata erros que ocorrem dentro de blocos protegidos. Para definir um tratamento de exceção, utiliza-se a seguinte construção :

try

      {comandos que se deseja proteger}

finally
      { comandos de tratamento de erros }

end;
A aplicação irá executar os comandos na parte except somente se ocorrer um erro. Se na parte  try chamar uma rotina que não trata erros, e um erro ocorrer, ao voltar para este bloco a parte  except será executada. Uma vez que a aplicação localiza um tratamento para a exceção ocorrida, os comandos são executados, e o objeto exceção é destruído. A execução continua até o fim do bloco.

Dentro da parte  except define-se um código a ser executado para manipular tipos específicos de exceção. Por exemplo, o código abaixo trata o erro de divisão por zero, através da exceção  Edivbyzero :

Function Divisao ( soma, numero : Integer) : integer;

Begin

Try

      Result:=  soma div numero;

Except
        On Edivbyzero  do  Result:= 0;

End;

A palavra reservada  ON define respostas para uma exceção. On está sempre junto de  do,  para manipular a exceção. Para ler informações específicas sobre o erro ocorrido, usa-se uma variação da estrutura  on..do, que provê uma variável temporária que engloba a exceção. Nesse caso, pode-se criar um quadro próprio de mensagens contendo a string da mensagem da exceção : 

Try

    {comandos}

Except
        On E : EInvalidOperation do  

         Messagedlg ( ‘Ignorando a Exceção : ‘ + E.Message, mtinformation , [mbok] , 0 );

End;

Onde a variável temporária  E é do tipo EinvalidOperation.

Pode-se prover um tratamento padrão de erro para tratar exceções que não tem tratamentos especificados. Para isto, adicione uma prate  else na parte except do bloco :

Try

    {comandos}

Except
        On EPrimeiroTipo do            

{código especificado para o primeiro tipo de erro}

        On EsegundoTipo  do           

{código especificado para o segundo tipo de erro}
        Else         

 {código padrão de tratamento de erros } 

End;

Além disso, pode-se utilizar as exceções genéricas para tratar um erro, em vez de uma exceção específica. Por exemplo , se desejar-se tratar um erro relacionado a uma operação com inteiros, mas não sabe exatamente o erro, poderá utilizar a exceção Einterror , que é a exceção genérica da qual se derivam outras exceções relacionadas a inteiros:

Try

     {Comandos}

except

        on EintError do 
       {Código de tratamento de erros}

end;
5. Exceções da VCL
Quando se escreve código chama rotinas da biblioteca de run-time ( RTL, run-time library), como funções matemáticas ou de manipulação de arquivos, os erros aparecem na forma de exceções. Por padrão, a RTL manda uma mensagem para o usuário, mas pode-se tratar estes erros de outra forma.

As exceções geradas pela RTL são definidas na unit  SysUtils , e todas descendem da classe mais geral Exception, que provê a string que aparece no quando de mensagem da exceção.

Há sete tipos de exceções geradas pela RTL :

· EDBEditError
Diz respeito a erros ocorridos com o uso do componente TDBEdit.

· EDataBaseError
Gerado sempre que ocorre um erro no Banco de Dados

· EDBEngineError
Gerado sempre ocorre um erro no BDE.

· EPrinter
Gerado quando ocorre um erro com a impressora

· EDDEError
Gerado a aplicação não consegue encontrar um servidor DDE

· EMCIDeviceError
Gerado quando ocorre um erro ao tentarmos acessar um dispositivo multimídia.

6. Exceções Silenciosas
Pode-se definir exceções que não mostrem um quadro de mensagem para o usuário quando aparecem. São chamadas exceções Sileciosas. O caminho mais curto para criar esta exceção é através da procedure Abort . Esta procedure automaticamente gera uma exceção do tipo Eabort , que abortará a operação sem mostrar uma mensagem. 

O exemplo abaixo aborta a operação de inclusão de itens em um Listbox quando tentamos inserir o terceiro elemento :

Begin

     For i := 1  to 10 do

            Begin
            Listbox.Items.Add ( Inttostr ( i ) );

            If i = 3  then
                       Abort;

             End;

End;
7. Manipulando Exceções Globais
O Delphi oferece um evento chamado OnException  , ligado à a classe Tapplication , que permite manipular qualquer exceção que ocorra em seu programa, mesmo que não sabendo em que parte do programa ela ocorreu. 

Inicialmente, deve-se criar manualmente, como um método de Tform1, a chamada para o método que tratará os erros :

Tform1 =  Class (Tform)

Procedure Trata_Erros (Sender : Tobject ; E : Exception );

End;

Depois de declarar a procedure, deve-se atribuí-la ao evento  OnException de Tapplication. A atribuição é feita no evento  OnCreate do formulário principal : 

Procedure Tform1.FormCreate ( Sender : Tobject );

Begin

Application.OnException := Trata_Erros;

End;

O método Trata_Erros será agora chamado quando qualquer exceção ocorrer. Pode-se determinar mensagens para erros específicos, ou mensagens gerais : 

Procedure Tform1.Trata_Erros (Sender : Tobject; E : Exception);

Begin

If  E is EdatabaseError then
        Showmessage(‘Erro no banco de dados’)

Else
         Showmessage( ‘Há erros no programa’);

End;

8. Traduzindo as Mensagens de Erro do Delphi

Se possuir um manipular de arquivos de recursos .RES (como o Resource WorkShop) , pode-se alterar as mensagens as mensagens de erros oferecidas pelo Delphi, traduzindo-as . Os arquivos de recursos do Delphi ficam no diretório \Arquivos de Programas\Borland\Delphi 3\Lib  

9 Tratando Uma Exceção

on <tipo de exceção> do

  <instrução>;

· Quando uma exceção é tratada, o fluxo de execução continua na instrução seguinte ao tratamento.

· Para permitir que o tratamento da exceção prossiga no tratamento default, utiliza-se a palavra-chave raise no fim do bloco de instruções do tratamento para que a exceção seja lançada novamente.

Raise

Object Pascal agora possui uma palavra reservada específica para exceções: raise. Raise possui dois diferentes usos: Primeiro, raise pode ser utilizado para re-gerar uma exceção que já tinha sido endereçada. Lembre-se de que uma vez que o bloco de manipulação de exceção tenha sido chamado para endereçar a exceção, o objeto exceção é destruído automaticamente. Em algumas situações você precisará manipular a exceção, mas ainda permitir que ela se propague para indicar à rotina chamadora que ocorreu uma exceção.

Eis um exemplo da utilização de Raise para re-gerar uma exceção. Se ocorrer um divisão por zero, você quer que a exceção se propague até a procedure chamadora. Entretanto, você também irá querer que a função nunca retorne um valor inválido. Raise permitirá que você manipule a exceção, alem de permitir que ela se propague.

function GetAverage( Sum : integer; NumItems : word ) : Double;

begin 

  try

    result := sum / NumItems;

  except

    on EDivByZero do

    begin

      result := 0.0;  // garantir o valor válido

      raise;          // indicar o erro ao chamador

    end; {EDivByZero}

  end; {except}

end;

Raise também pode ser utilizado para gerar manualmente uma exceção, simulando uma condição de erro. Se você quisesse gerar uma exceção EDivByZero por si próprio, você poderia fazê-lo com a seguinte sintaxe:

raise EDivByZero.create( ‘Division by zero’ );   

Raise pode ser utilizado desta forma para chamar manualmente o construtor para uma exceção. O parâmetro para o construtor é a mensagem default que será exibida pelo objeto application. Entretanto, mesmo que você chame o construtor com raise, você não deve chamar o destrutor. O Delphi sempre chama destrutores para exceções, não importando como foram gerados. Raise também pode ser utilizado desta forma para construir exceções definidas pelo usuário, como veremos em uma seção mais adiante. Raise também tem um papel importante no capítulo a seguir, sobre manipulação avançada de exceção, particularmente para erros de engine de banco de dados.

if <condição anormal> then

  raise <classe de exceção >.Create( <descrição> );

11) Criação de Classes de Exceção

<identificador> = class(Exception);

· Qualquer anormalidade na execução de um programa pode ser transformada em uma exceção pela criação de uma classe para fazer esse tratamento.

· As classes de exceção devem ser subclasses da classe Exception.

· No momento em que houver uma anormalidade, um objeto do classe de exceção criada deve ser lançado, seguindo o modelo:

Exemplos:

type

  EFatorialNegativo = class(Exception)

  public

    constructor Create;

  end;

constructor EFatorialNegativo.Create;

begin

  inherited Create('Fatorial Negativo');

end;

function Fatorial(const n: Longint): Longint;

begin

  if n < 0 then

    raise EFatorialNegativo.Create;

  if n = 0 then

    Result := 1

  else

    Result := n * Fatorial(n - 1);

end;

var

  N: Longint;

begin

  Write('Digite um valor (-1 para sair): ');

  Read(N);

  while N <> -1 do

  begin

    try

      Writeln(n, '! = ', Fatorial(N));

    except

      on E:EFatorialNegativo do

        Writeln(E.Message);

    end;

    Write('Digite um valor (-1 para sair): ');

    Read(N);

  end;

end.

Exemplos:

var

  F : File;

begin

  AssignFile(F,’Arquivo.xxx’);

  Reset(F);

  try

    { código que atualiza o arquivo F}

  finally

    CloseFile(F);

  end;

end;

Principais Classes de Exceção

	Classe
	Descrição

	EaceessViolation
	Acesso inválido a uma região de memória.

	EConvertError
	As funções StrToInt ou StrToFloat não consegue converte uma string num valor numérico válido

	EDivByZero
	Divisão de inteiro por zero

	EGPFault
	Acesso a memória não permitido

	EInOutError
	Erro de I/O de arquivo

	EIntOverFlow
	Operação inválida com número inteiro

	EinvalidCast
	Typecast inválido utilizando o operado As

	EinvalidOp
	Operação inválida com número real

	EoutOfMemory
	Memória Livre insuficiente para alocar objeto

	Eoverflow
	Número real excedeu a faixa válida

	ErangeError
	Valor excede a faixa definida para o inteiro

	EstackOverflow
	Quando a pilha (stack) não pode crescer dinamicamente

	Eunderflow
	Número real menor que a faixa válida

	EzeroDivide
	Divisão de real por zero


Aplicações Robustas

O tratamento de exceção é um mecanismo capaz de dar robustez a uma aplicação, permitindo que os erros sejam manipulados de uma maneira consistente e fazendo com que a aplicação possa se recuperar de erros, se possível, ou finalizar a execução quando necessário, sem perda de dados ou recursos.

Para que uma aplicação seja segura, seu código necessita reconhecer uma exceção quando esta ocorrer e responder adequadamente a essa exceção. Se não houver tratamento para uma exceção, será exibida uma mensagem padrão descrevendo o erro e todos os processamentos pendentes não serão executados. Uma exceção deve ser respondida sempre que houver perigo de perda de dados ou de recursos do sistema.

Exceções

Exceções são classes definidas pelo Delphi para o tratamento de erros. Quando uma exceção é criada, todos os procedimentos pendentes são cancelados e, geralmente é mostrada uma mensagem de erro para o usuário. As mensagens padrão nem sempre são claras, por isso é indicado criar seus próprios blocos protegidos.

Blocos Protegidos

Um bloco protegido é um grupo de comandos com uma seção de tratamento de exceções.

try

  A := StrToFloat(EdtA.Text);

  B := StrToFloat(EdtB.Text);

  ShowMessage(Format('%f / %f = %f', [A, B, A + B]));

except

  ShowMessage('Números inválidos.');

end;
Algumas vezes você pode precisar especificar quais exceções quer tratar, como mostrado abaixo.

try

  Soma := StrToFloat(EdtSoma.Text);

  NumAlunos := StrToInt(EdtNum.Text);

  ShowMessage(Format('Média igual a %f.', [Soma / NumAlunos]));

except

  on EConvertError do

    ShowMessage('Valor inválido para soma ou número de alunos.');

  on EZeroDivide do

    ShowMessage('O número de alunos tem que ser maior que zero.');

  else

    ShowMessage('Erro na operação, verifique os valores digitados.');

end;

Principais Exceções

O Delphi define muitas exceções, para cada erro existe uma exceção correspondente.

	Classe
	Descrição

	Exception
	Exceção genérica, usada apenas como ancestral de todas as outras exceções

	EAbort
	Exceção silenciosa, pode ser gerada pelo procedimento Abort e não mostra nenhuma mensagem

	EAccessViolation
	Acesso inválido à memória, geralmente ocorre com objetos não inicializados

	EConvertError
	Erro de conversão de tipos

	EDivByZero
	Divisão de inteiro por zero

	EInOutError
	Erro de Entrada ou Saída reportado pelo sistema operacional

	EIntOverFlow
	Resultado de um cálculo inteiro excedeu o limite

	EInvalidCast
	TypeCast inválido com o operador as

	EInvalidOp
	Operação inválida com número de ponto flutuante

	EOutOfMemory
	Memória insuficiente

	EOverflow
	Resultado de um cálculo com número real excedeu o limite

	ERangeError
	Valor excede o limite do tipo inteiro ao qual foi atribuída

	EUnderflow
	Resultado de um cálculo com número real é menor que a faixa válida

	EVariantError
	Erro em operação com variant

	EZeroDivide
	Divisão de real por zero

	EDatabaseError
	Erro genérico de banco de dados, geralmente não é usado diretamente

	EDBEngineError
	Erro da BDE, descende de EDatabaseError e traz dados que podem identificar o erro


Blocos de Finalização

Blocos de finalização são executados sempre, haja ou não uma exceção. Geralmente os blocos de finalização são usados para liberar recursos.

FrmSobre := TFrmSobre.Create(Application);

try

  FrmSobre.Img.LoadFromFile('Delphi.bmp');

  FrmSobre.ShowModal;

finally

  FrmSobre.Release;

end;

Você pode usar blocos de proteção e finalização aninhados

FrmOptions := TFrmOptions.Create(Application);

try

  FrmOptions.ShowModal;

  try

    Tbl.Edit;

    TblValor.AsString := EdtValor.Text;

  except

    on EDBEngineError do 

      ShowMessage('Alteração não permitida.');

    on EConvertError do 

      ShowMessage('Valor inválido.');

  end;

finally

  FrmOptions.Release;

end;

Geração de Exceções

Você pode provocar uma exceção usando a cláusula raise.

raise EDatabaseError.Create('Erro ao alterar registro.');

Também é possível criar seus próprios tipos de exceções.

type

  EInvalidUser = class (Exception);

raise EInvalidUser.Create('Você não tem acesso a essa operação.');

Se você quiser que uma exceção continue ativa, mesmo depois de tratada, use a cláusula raise dentro do bloco de tratamento da exceção. Geralmente isso é feito com exceções aninhadas.

try

  Tbl.Edit;

  TblContador.Value := TblContador.Value + 1;

  Tbl.Post;

except

  ShowMessage('Erro ao alterar contador.');

  raise;

end;

Erros de Bancos de Dados

A exceção EDBEngineError permite a identificação de erros de bancos de dados gerados pela BDE.

try

  TblCli.Post;

except

  on E: EDBEngineError do 

    if E.Errors[0].ErrorCode = DBIERR_KEYVIOL then

      ShowMessage('Cliente já cadastrado.');

end;

Note que a variável E, que vai identificar o erro, só precisa ser declarada no bloco de tratamento da exceção. No help você pode consultar outras propriedades de EDBEngineError que podem ser importantes.

Você também pode usar os eventos de erro do componente Table, sem precisar de blocos de tratamento.

procedure TFrmCadCli.TblCliPostError(DataSet: TDataSet; E: EDatabaseError;

  var Action: TDataAction);

begin

  if(E is EDBEngineError) then

    with EDBEngineError(E) do

      case Errors[0].ErrorCode of

        DBIERR_KEYVIOL: ShowMessage('Cliente já cadastrado.');

        DBIERR_REQDERR: ShowMessage('Campo obrigatório não preenchido.');

      end

  else

    ShowMessage('Erro no banco de dados:' + #13#13 + E.Message);

  Action := daAbort;

end;
	Constante
	Descrição

	DBIERR_KEYVIOL
	Violação de chave primária

	DBIERR_MAXVALERR
	Valor máximo excedido

	DBIERR_FORIEGNKEYERR
	Erro de chave externa, como em integridade referencial

	DBIERR_LOCKED
	Registro travado

	DBIERR_FILELOCKED
	Arquivo travado

	DBIERR_NETMULTIPLE
	Mais de um diretório usado como NetFileDir

	DBIERR_MINVALERR
	Campo com valor mais baixo que valor mínimo

	DBIERR_REQDERR
	Campo obrigatório faltando

	DBIERR_LOOKUPTABLEERR
	Erro em tabela Lookup


Se você quiser mais informações a respeito do erro pode usar o procedimento DBIGetErrorContext, como na função mostrada abaixo que retorna determinadas informações sobre o erro.

function GetErrorInfo(Context: SmallInt): string;

begin

  SetLength(Result, DBIMAXMSGLEN + 1);

  try

    DbiGetErrorContext(Context, PChar(Result));

    SetLength(Result, StrLen(PChar(Result)));

  except

    Result := '';

  end;

end;

No evento OnEditError, usado no exemplo abaixo, se ocorrer um erro ao tentar alterar um registro, podemos identificar o usuário da rede que está alterando esse registro usando a função criada anteriormente.

if Pos('locked', E.Message) > 0 then

  ShowMessage('Usuário ''' + GetErrorInfo(ecUSERNAME) + ''' está alterando o registro.');

Note que foi usada uma outra técnica de identificação do erro, usando a própria mensagem de erro e não o código, como mostrado anteriormente. Você pode usar a função criada acima mandando como parâmetro os valores mostrados abaixo, que podem ser encontrados no help da BDE.

	Constante
	Descrição

	ecTABLENAME 
	Nome da Tabela

	ecFIELDNAME 
	Nome do campo

	ecUSERNAME 
	Nome do usuário, muito usado para identificar qual usuário travou o registro

	ecFILENAME 
	Nome do arquivo

	ecINDEXNAME 
	Nome do índice

	ecDIRNAME 
	Pasta

	ecKEYNAME 
	Chave primária

	ecALIAS 
	Alias

	ecDRIVENAME 
	Drive

	ecNATIVECODE
	Código de erro nativo

	ecNATIVEMSG
	Mensagem de erro nativa

	ecLINENUMBER
	Número da linha, usado em instruções SQL


Para desenvolver um sistema genérico de tratamento de erros, considere a opção de criar esse tratamento em um DataModule genérico para ser usado como ancestral por todos os DataModules do sistema, utilizando a herança visual.

Se o único problema for traduzir as mensagens, localize os arquivos CONSTS.INT e DBCONSTS.INT e crie uma nova Unit de definição de strings com uma estrutura semelhante a mostrada abaixo e juntando todas as definições das constantes das duas Units devidamente traduzidas. Depois, basta usar essa Unit em seus projetos que as novas mensagens irão sobrepor as anteriores.

unit NewConsts;

interface

resourcestring

  SAssignError = 'Não é possível atribuir %s a %s';

  SFCreateError = 'Não é possível criar arquivo %s';

  SFOpenError = 'Não é possível abrir arquivo %s';

  SInvalidFieldSize = 'Tamanho de campo inválido';

  SInvalidFieldRegistration = 'Registro de campo inválido';

  SUnknownFieldType = 'Campo ''%s'' tem um tipo desconhecido';

implementation

end.

Uma outra opção seria criar um método para o evento OnException do objeto Application, esse método seria chamado sempre que houvesse uma exceção em qualquer parte do sistema.

















